Notes From Vector Calculus

1. A vector field is a function from $\mathbb{R}^2$ to $\mathbb{R}^2$ that typically represents some force. For example, plot the vector field:

   $$\mathbf{F}(x,y) = \langle \ln(1+y^2), \ln(1+x^2) \rangle$$

   ```
   with(plots):
   fieldplot([ln(1+y^2),ln(1+x^2)],x=-5..5,y=-5..5);
   ```

   Example for you: Plot the vector field

   $$\langle \frac{x-2y}{\sqrt{1+x^2+y^2}}, \frac{x-2}{\sqrt{1+x^2+y^2}} \rangle$$

   Plot it as `fieldplot[P,Q]`, where P and Q are Maple variables that you have defined appropriately.

2. Definition: A vector field is said to be conservative if it came from the gradient of a function:

   $$\mathbf{F}(x,y) = \langle f_x, f_y \rangle$$

   One way to see if you have a conservative vector field is to invoke Clairaut's Theorem. That is, if the vector field is $\langle P, Q \rangle$, check if $P_y = Q_x$. In that case, we think of $f_{xy} = P_y = Q_x = f_{yx}$, and we can antidifferentiate to find the source function f.

   Example for you: Is the following vector field conservative? If so, find the f so that this is the gradient:

   $$\langle 3\cos(3x-2y), -2\cos(3x-2y) \rangle$$

3. The streamlines (or flow lines) of a vector field are the paths followed by a particle whose velocity field is the given vector field.

   In such a case, if the vector field is $\langle P, Q \rangle$, then the path that the particle takes is defined by the differential equations:

   $$\frac{dx}{dt} = P(x,y) \qquad \frac{dy}{dt} = Q(x,y)$$

   Side Remark: If you've taken differential equations, you might recognize this as a system of autonomous first order DEs. If the vector field is conservative, we called this ODE *exact*, and the differential equation for it would be:

   $$\frac{dy}{dx} = \frac{\frac{dy}{dt}}{\frac{dx}{dt}} = \frac{Q}{P} \quad \Rightarrow \quad P\,dy = Q\,dx \quad \Rightarrow \quad -Q\,dx + P\,dy = 0$$

   Example: Numerically solve the given differential equations if P, Q are as in Note 1 above.

```
with(plots):
with(DEtools):
P:=ln(1+y(t)^2); Q:=ln(1+x(t)^2);
sys:={diff(x(t),t)=P,diff(y(t),t)=Q};
A:=DEplot(sys,[x(t),y(t)],t=0..4,[[x(0)=-4,y(0)=0],
   [x(0)=-4,y(0)=-5],[x(0)=-4,y(0)=-4.1]],x=-5..5,
   y=-5..5,linecolor=black):
B:=fieldplot([ln(1+y^2),ln(1+x^2)],x=-5..5,y=-5..5):
display({A,B});
```

What should you see? You should see the force field and two solution curves to the differential equations- following the forces in the field.

4. We can connect this idea to another- Finding the maximum of a function. We know that, given a surface like:

$$z = f(x, y)$$

and a point (a, b), then to increase f the fastest, we should move in the direction of the gradient. We can translate this into a system of differential equations:

$$\frac{dx}{dt} = f_x(x, y) \qquad \frac{dy}{dt} = f_y(x, y) \qquad x(0) = 1, y(0) = b$$

The commands we would use to plot this would be similar to the ones given.

We also know that the gradient is perpendicular to the contours of the function. In that case, we could create a set of differential equations for that as well:

$$\frac{dx}{dt} = -f_y(x, y) \qquad \frac{dy}{dt} = f_x(x, y)$$

Example: Let $z = f(x, y) = 1 - (x^2 + y^2)$. Then:

$$\nabla f = \langle -2x, -2y \rangle$$

Notice that for any point, this direction is pointed towards the origin, which is where the maximum of this function occurs. In fact, the gradient tells us the quickest way to get to the maximum (in terms of the rate of change of f). If we wanted to travel to the top, the differential equations would be:

$$\frac{dx}{dt} = -2x \qquad \frac{dy}{dt} = -2y$$

On the other hand, the contours are perpendicular to the gradient, and the differential equations that correspond with contours would be:

$$\frac{dx}{dt} = 2y \qquad \frac{dy}{dt} = -2x$$

In the Maple example below, we show the orthogonal sets of vectors. Try changing this example to travel to the maximum of the function.

```
f:=1-(x^2+y^2);
dfx:=diff(f,x);
dfy:=diff(f,y);

with(plots):
with(DEtools):
A:=fieldplot([dfx,dfy],x=-5..5,y=-5..5):
Eqn1:=diff(x(t),t)=subs(x=x(t),y=y(t),-1*dfy);
Eqn2:=diff(y(t),t)=subs(x=x(t),y=y(t),dfx);
B:=DEplot([Eqn1,Eqn2],[x(t),y(t)], t=0..8,
   x=-5..5,y=-5..5,[[x(0)=1,y(0)=-1],
   [x(0)=1,y(0)=1.5]],stepsize=0.01):

display({A,B});
```

5. Line integrals and vector fields: To compute a line integral, we need to parameterize a path within some vector field. For example, in three dimensions we might have something like the following (see if you can understand the Maple code):

```
restart;
Velocity:=<(-y-z)/(x^2+y^2), (x-z)/(x^2+y^2),(-z^2*(x+y))/(x^2+y^2)^2>;

x:=cos(t); y:=sin(t); z:=4;
r:=<x,y,z>;
with(VectorCalculus):
rprime:=diff(r,t);
Circulation:=Int(DotProduct(Velocity,rprime),t=0..2*Pi);
evalf(Circulation)
```

Notice that in defining the path *after* the velocity, the x, y, z were automatically substituted in the Velocity expression when we performed the integration. In computing the line integral where the path C is parameterized by $\mathbf{r}(t)$, we are computing:

$$\int_C \mathbf{F} \cdot d\mathbf{r} \doteq \int_a^b \mathbf{F}(\mathbf{r}(t)) \cdot \mathbf{r}'(t)\, dt$$

If we're moving along the path C in the force field, then the line integral represent the work involved. If $\mathbf{F} = \langle P, Q \rangle$, then the left side of the equation is sometimes written as:

$$\int_C P\, dx + Q\, dy$$

6. Connecting Line Integrals to Green's Theorem:

$$\int_C P\, dx + Q\, dy = \int \int_D \left(\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} \right)\, dA$$

We'll look at a particular example:

Suppose we have a vector field $\mathbf{F} = \langle y^2, 3xy \rangle = \langle P, Q \rangle$. The path C is (clockwise) along the upper half of an annulus, bounded by:

$$x > 0, \quad x^2 + y^2 \leq 4, \quad x^2 + y^2 \geq 1$$

In polar coordinates, this is particularly easy to represent:

$$\{(r, \theta) \mid 1 \leq r \leq 2, \quad 0 \leq \theta \leq \pi\}$$

We will create a parameterization and evaluate the integral directly, then compare to the result of Green's Theorem.

(a) The path is broken into 4 parts:

i. $x(t) = 2\cos(t), y(t) = 2\sin(t)$, $t \in [0, \pi]$
ii. $x(t) = 2 - t$, $y(t) = 0$, $t \in [0, 1]$
iii. $x(t) = \cos(t), y(t) = \sin(t)$, t from π to 0
iv. $x(t) = 1 + t$, $y(t) = 0$, $t \in [0, 1]$

Paths 2 and 4 will be zero (since $y = 0$, verify). Here are the Maple commands, in detail:

```
restart;
P:=y^2; Q:=3*x*y;
x:=2*cos(t); y:=2*sin(t); dx:=diff(x,t); dy:=diff(y,t);
P1:=int(P*dx+Q*dy,t=0..Pi);
x:=2-t; y:=0; dx:=diff(x,t); dy:=diff(y,t);
P2:=int(P*dx+Q*dy,t=0..1);
x:=cos(t); y:=sin(t); dx:=diff(x,t); dy:=diff(y,t);
P3:=int(P*dx+Q*dy,t=Pi..0);
```

(b) On the other hand,

$$\frac{\partial Q}{\partial x} - \frac{\partial P}{\partial y} = 3y - 2y = y$$

and in polar coordinates, $y = r\sin(\theta)$, so the line integral becomes:

$$\int_0^\pi \int_1^2 r\sin(\theta)\, r\, dr\, d\theta = \frac{14}{3}$$

7. Final Example for today: Line integrals can be used to compute the area within simple regions (there is a mathematical definition of *simple*- Try to look it up!):

$$\int_C x\, dy = -\int_C y\, dx$$

Where C is a path along the boundary (going counterclockwise). Verify these formulas for the ellipse defined by:

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$$

The parameterization for an ellipse is $x = a\,\cos(t)$, $y = b\sin(t)$, therefore:

$$-\int_C y\,dx = -\int_0^{2\pi} ab\sin(t)(-\sin(t))\,dt = \pi ab$$

Exercises:

1. Find the work done by **F** over the given path:

$$\mathbf{F} = \langle y + yz\cos(xyz), x^2 + xz\cos(xyz), z + xy\cos(xyz)\rangle$$

   where the path is on the ellipse, $\frac{x^2}{4} + \frac{y^2}{9} = 1$ and the height is fixed at $z = 1$. Does it matter which direction you take (Check!)

2. Let $f(x, y) = x\mathrm{e}^{-x^2-y^2}$:

   (a) Plot the surface with $-2 \leq x \leq 2$, same interval for y.

   (b) Plot some contours for the surface (you can use Maple's default values).

   (c) Starting at the point $(0, 3/2)$, find the best path to the top of hill using differential equations. Plot this path together with the contours you found earlier.

   (d) How would you find a path to the bottom of the valley? Try it, and plot that result together with the previous result.

3. Astroids[1]!

   (a) An astroid is defined by the parametric equations $x = \cos^3(t), y = \sin^3(t)$, where t runs from 0 to 2π. Plot this astroid in Maple.

   (b) If $h(x, y) = x^3y^5$, compute the line integral of h over the part of the astroid in the first quadrant. Pause for a moment to reflect on how nasty it would be to grind through this problem by hand.

   (c) Find the area of the astroid.

[1] No, not asteroids...